

List Processing Macro Function CallText: Read a Control Data Set and Unquote each Item

Ronald J. Fehd, Stakana Analytics

Abstract **Description:** The macro language of SAS® software provides the %nrstr function which can be used to delay resolution of macro variable references until execution time, at which time the %unquote function is used to call for the resolution of reference. This paper shows the issues in writing a macro function which reads a list, a control data set, and returns each item surrounded by user-provided special characters. This macro uses scl functions to read the data set thus avoiding the necessity of a data step.

Purpose: The purpose of this paper is to provide a working example of a macro function which calls explicitly state the special characters desired to enclose each item in the list.

Audience: programmers writing list processing applications

Keywords: hexadecimal, functions: datetime, getoption, putn;
macro functions: %nrstr, %sysfunc, %unquote;
scl functions: attrn, close, exist, fetchobs, getvarc, getvarn, open, varname, varnum, vartype

In this paper	Introduction	1
	Programs, Log and Listing	3
	Summary	8
	References	9

Introduction

Overview The section contains these topics.

- description
- parameters
- algorithm

description Macro `calltext.sas` is a list-processing function and routine which reads a control data set, converts each variable and its value in each observation into macro variables and then unquotes the parameter `text` which contains macro variable references as well as quotes and other special characters.

parameters

This is the list of parameters.

- `data`, one- or two-level data set name
- `text`, text with macro variable references and special characters
- `delimiter`, text to add between expansion of parameter `text`
- `global`, boolean: write macro variables to global symbol table;
i.e.: make macro variables from row available to other processes
- `hex16`, boolean: convert numerics to hexadecimal

algorithm

This is the list of steps in the algorithm.

1. assertion: data exists
2. fetch data information: n-obs and n-vars
3. assertion: data has n-obs and n-vars
4. for each observation:
 - (a) for each variable:
 - i. fetch variable information: name, number and type
 - ii. allocate macro variable
 - (b) return the string contained in the parameter `text`
with macro variable references expanded

hard-coded example

This program shows the desired results of the function.

```

1  *function returns tokens within DATA statement;
2  DATA work.class_sex_ F
3      work.class_sex_ M
4      ;
5  do until(endofile);
6      set sashelp.class end = endofile;
7      *returns statements and delimiter;
8      if sex eq ' F ' then output
9      work.class_sex_ F;
10     /*delimiter; else
11     if sex eq ' M ' then output
12     work.class_sex_ M;
13     end;
14 stop;
15 run;

```

Programs, Log and Listing

Overview

This is the list of programs, logs and listings shown here.

- macro calltext.sas
 - program calltext-test.sas
 - subroutine proc print with titles
 - calltext-test.log
 - calltext-test.lst
-

macro calltext

```

1  /* name: <UNC>\SAS-site\macros\calltext.sas
2  author: Ronald J. Fehd 2012
3  -----
4  Summary : description  : call text with specified
5             values in data set as parameters
6             purpose     : provide generic method to call text
7             using list==control data set of parms
8  -----
9  Contexts: program group: list processing token generator
10             program type: function
11             SAS      type: macro assignment statement generator
12             uses routines: n/a
13  -----
14  Specifications: input  : required: data, text
15                     optional: delimiter, hex16
16                     process: assemble macro variable assignment(s)
17                     call
18                     output : from text
19  Parameters: data      = one- or two-level data set name
20             ,text     = text of macro references for each row
21             *-constraint-*: must be enclosed in nrstr:
22             ,text     = %nrstr(%put note: name:&name sex=&sex;)
23             notes: may be any of:
24             * tokens
25             * statements including semicolons
26             * macro calls:
27             ,text     = %nrstr(%mymacro(name=&name,sex=&sex))
28             ,hex16    = 1 :: default, convert numeric to hex16
29                     used to pass real numbers accurately
30                     across step boundaries
31             ,hex16    = 0 :: pass numerics as decimals
32             ,testing  = 0 :: default, no extra messages in log
33             ,testing  = 1 :: note: auto-reset when
34                     options mprint source2;
35             ,unquote  = 1 :: default, write notes to log
36             ,unquote  = 0 :: do not   write notes to log
37  -----
38  bells, whistles: writes real time used to log
39                   note: CALLTEXT used real time  0:00:00.016
40  -----

```

```

41  usage example:
42  PROC freq data  = sashelp.class;
43      tables  sex / noprint
44      out = freq_class_sex;
45  run;*necessary;
46  DATA %calltext(data = freq_class_sex
47      ,text = %nrstr(work.sex_&sex)
48      );*end of data statement;
49  do until(endofile);
50      set sashelp.class  end = endofile;
51      %calltext(data      = freq_class_sex
52      ,delimiter = %nrstr(else)
53      ,text
54      =%nrstr(if sex eq "&sex" then output work.sex_&sex;)
55      )
56  end;
57  stop;
58  run;
59  log: -----
60      13      DATA
61  MPRINT(CALLTEXT):  work.sex_F
62  MPRINT(CALLTEXT):  work.sex_M
63      18      do until(EndoFile);
64      19      set sashelp.class end = endoFile;
65  MPRINT(CALLTEXT): if Sex eq "F" then output work.sex_F;
66  MPRINT(CALLTEXT): else
67  MPRINT(CALLTEXT): if Sex eq "M" then output work.sex_M;
68      25      end;
69      26      stop;
70      27      run;
71  NOTE: There were 19 obs read from data set SASHELP.CLASS
72  NOTE: The data set WORK.sex_F has 9 obs and 5 variables
73  NOTE: The data set WORK.sex_M has 10 obs and 5 variables
74  -----
75  DateTime: 2016-09-09 7:10:13 AM
76  word count:      words: 691
77                  lines: 157
78  characters(no spaces): 4591
79  characters(with spaces): 6584
80  ***** ..... */
81  %MACRO calltext
82      (data      = sashelp.class
83      ,text      = %nrstr(%put note: name:&name sex=&sex;)
84      ,delimiter= /* %nrstr(else) */
85      ,global    = 0 /* make mvars global? */
86      ,hex16     = 1 /* convert numerics to hex16? */
87      ,testing   = 0
88      ,unquote   = 1 /* write notes to log? */
89  )/ des =
90  'site: assemble mvar assignment statement(s), unquote text'
91  /** * store source /** */
92  ;/**** NOTE: _*: avoid name collisions w/data set vars ****/
93  %local _delimiter _global _hex16 _text _testing _unquote
94      _dsid _rc _obs _n_obs _var _n_vars
95      _name _num _type _time_start _time_end;
96  %let _testing = %eval(not(0 eq &testing)

```

```

97         or( %sysfunc(getoption(mprint)) eq MPRINT
98           and %sysfunc(getoption(source2)) eq SOURCE2));
99 %let _time_start= %sysfunc(datetime(),hex16);
100 %let _delimiter = &delimiter;
101 %let _global    = %eval(not(0 eq &global));
102 %let _hex16     = %eval(not(0 eq &hex16));
103 %let _text      = &text;
104 %let _unquote   = %eval(not(0 eq &unquote));
105
106 /** description: assertions;
107 ** purpose   : if fail then exit;
108 %if not(%sysfunc(exist(&data))) %then %do;
109     %put Err%str()or &sysmacroname: not exist(&data);
110     %return;
111 %end;
112 %let _dsid = %sysfunc(open (&data          ));
113 %let _n_obs = %sysfunc(attrn(&dsid,nobs ));
114 %let _n_vars = %sysfunc(attrn(&dsid,nvars));
115 %put info: &sysmacroname &=data &=_n_obs &=_n_vars;
116 %if not &n_obs or not &n_vars %then %do;
117     %put Err%str()or &sysmacroname exit: not(nobs and vars);
118     %goto close_exit;
119 %end;
120
121 /** description: read data;
122 ** purpose   : make macro variable assignment(s), submit;
123 %do _obs = 1 %to &n_obs;**** fetchobs==read row;
124     %let _rc = %sysfunc(fetchobs(&dsid,&_obs ));
125     %do _var = 1 %to &n_vars;
126         %let _name = %sysfunc(varname (&dsid,&_var ));
127         %let _num = %sysfunc(varnum (&dsid,&_name));
128         %let _type = %sysfunc(vartype (&dsid,&_num ));
129         %if &_global %then %global &_name;
130         /** assign: for type=C: value, type=N: hex16(value);
131         %if &_type eq C
132             or(&_type eq N and not &_hex16) %then
133                 %let &_name = %left(%sysfunc(
134                     getvar&_type(&dsid,&_num)));
135             %else %let &_name = %left(%sysfunc(putn(
136                 %sysfunc(getvar&_type(&dsid,&_num)),hex16)));
137             %if &_testing or &_global %then
138                 %put echo: &_name=&&&_name;
139             %end;
140
141         %if not &_testing and &_unquote %then
142             %put &sysmacroname output: %cmpres(%unquote(&_text));
143         /** submit: note! no ending semicolon!;
144         %unquote(&_text)
145
146         %if %length(&_delimiter) and &_obs lt &n_obs %then %do;
147             %if not &_testing and &_unquote %then
148                 %put &sysmacroname output: %unquote(&_delimiter);
149             %unquote(&_delimiter)
150             %end;
151         %end;
152 %if &_testing %then %put _local_;

```

```

153
154 %close_exit: %let _rc      = %sysfunc(close(&_dsid));
155                %let _time_end = %sysfunc(datetime(),hex16);
156 %put info: &sysmacroname used real time %sysfunc
157            (putn(&_time_end.x-&_time_start.x,time12.3));
158 %mend calltext;

```

program
calltext-test.sas

```

1  * name: call-text-test-class-print.sas;
2  options mprint;* source2; * testing on;
3  %let name = sex;
4  PROC freq data  = sashelp.class;
5      tables    &name / noprint
6      out = freq_class_&name;
7  run;*necessary;
8  DATA %calltext(data = freq_class_&name
9      ,text = %nrstr(work.&name._&&&name)
10     );*end of data statement;
11 do until(endofile);
12     set sashelp.class end = endofile;
13     %calltext(data      = freq_class_&name
14     ,delimiter = %nrstr(else)
15     ,text=%nrstr
16     (if &name eq "&&&name" then output work.&name._&&&name;)
17     )
18     end;
19 stop;
20 run;
21 options nomprint nosource2; * testing off;
22 *replicates method of CallXinc::cxinclude;
23 %let sex    =.;
24 %let count  =.;
25 %let percent=.;
26 %calltext(data      = freq_class_&name
27     ,text      = %nrstr
28     (%include project(proc-print-subset-w-titles);)
29     ,global     = 1)
30 %put &=sex &=count &=percent;

```

log

```

1  8      DATA %calltext(data = freq_class_sex
2  9                          ,text = %nrstr(work.sex_&sex)
3  10                          );*end of data statement;
4  MPRINT(CALLTEXT):  work.sex_F
5  MPRINT(CALLTEXT):  work.sex_M
6  11      do until(endofile);
7  12          set sashelp.class end = endofile;
8  13          %calltext(data      = freq_class_sex
9  14                          ,delimiter = %nrstr(else)
10 15                          ,text
11 16                          =%nrstr(if sex eq "&sex" then output work.sex_&sex;)
12 17                          )
13 MPRINT(CALLTEXT): if sex eq "F" then output work.sex_F;
14 MPRINT(CALLTEXT): else
15 MPRINT(CALLTEXT): if sex eq "M" then output work.sex_M;
16 18          end;
17 19          stop;
18 20          run;
19 NOTE: There were 19 obs read from the data set SASHELP.CLASS.
20 NOTE: The data set WORK.SEX_F has 9 observations and 5 variables
21 NOTE: The data set WORK.SEX_M has 10 observations and 5 variables
22 info: CALLTEXT DATA=freq_class_sex _N_OBS=2 _N_VARS=3
23 CALLTEXT output: %include project(proc-print-subset-w-titles);
24 echo parameters NAME=sex: F COUNT  =4022000000000000
25                          PERCENT=4047AF286BCA1AE7

```

routine proc print with titles

This routine shows how to format macro variables saved as hex16 back to integer and real numbers.

```

1  *name: proc-print-subset-w-titles.sas;
2  *description: shows how to convert numbers in mvars saved as hex16
3                into integers and reals;
4  %put echo parameters &=name: &&&name &=count &=percent;
5  PROC print data  = work.&name._&&&name;
6                Title2 "data      : work.&name._&&&name";
7                Title3 "subset   : &name eq &&&name";
8                Title4 "freq     : count   : &count";
9                Title5 "         : percent: &percent";
10               Title6 "         count: %left(%sysfunc(putn(&count.x ,32. )))";
11               Title7 "         percent: %left(%sysfunc(putn(&percent.x,32.16)))";
12               Title8 "         rounded: %left(%sysfunc(putn(&percent.x,6.2 )))";
13 run;

```

partial listing

List Processing macro function CallText

```

data      : work.sex_F
subset    : sex eq F
freq      : count   : 4022000000000000
           : percent: 4047AF286BCA1AE7
           count: 9
           percent: 47.0000000000000000
           rounded: 47.00

```

Obs	Name	Sex	Age	Height	Weight
1	Alice	F	13	56.5	84.0
2	Barbara	F	13	65.3	98.0

Summary

Suggested Reading

This function is a sibling of list processing macro function call-macro which is described in Fehd [5]. Both `call-macro` and `call-text` are derivative works of the list processing routine CallXinc described in Fehd [4], which is a derivative work of the SmryEachVar suite Fehd [3].

- list processing : Fehd and Carpenter [6] provides an overview of the issues of creating and processing lists of macro variables
 - macro processor : Lavery [7] provides a definitive overview of the interaction of the tokenization of macro and SAS statements; Chaudhary [1] discusses differences between macro compile and execution functions; Russell [11], author from SAS Tech Support provides answers to top 10 FAQ (Frequently Asked Questions); Lee [8] compares macro functions `%nrstr` and `%nrquote`; Rosenbloom and Carpenter [10] shows how to store and retrieve special characters in macro variables;
 - macros for lists : Clay [2] discusses vertical arrays of macro variables and provides examples of macro functions; Tsai [12] discusses use of horizontal and vertical arrays of macro variables; Whitlock [13], an acknowledged master of macro usage, discusses use of quotes and separators when writing macro functions which process lists of tokens; Morris [9] shows common problems in macros for list processing
-

Conclusion

SAS Component Language (SCL) functions can be used to open a data set, read each row, and, within each row, fetch variable information which is used to allocate macro variables; with a list of macro variables for each row the parameter `text` can then be expanded returning strings which may be tokens within a SAS statement or one or more statements. The use of the pair of macro functions, `%nrstr` during compilation and `%unquote` during execution, allows a programmer to write code which is easily understood.

Author Information

Ronald J. Fehd

Ron.Fehd.macro.maven@gmail.com

About the author:

sco.wiki

http://www.sascommunity.org/wiki/Ronald_J._Fehd

LinkedIn

www.linkedin.com/Ronald.Fehd

affiliation

Stakana Analytics, Senior Maverick

also known as

macro maven on SAS-L, Theoretical Programmer

Programs:

<http://www.sascommunity.org/wiki/>

Macro_CallText

Trademarks

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. In the USA and other countries ® indicates USA registration. Other brand and product names are trademarks of their respective companies.

References

- [1] Kaushal Raj Chaudhary. Essentials of macro quoting functions in SAS(R). In *MidWest SAS Users Group Annual Conference Proceedings*, 2015. URL <http://www.mwsug.org/proceedings/2015/RF/MWSUG-2015-RF-08.pdf>. Rapid Fire, 4 pp., introduction to macro quoting functions, discusses difference between compile (assignment) and execution (resolution) functions.
 - [2] Ted Clay. Tight looping with macro arrays. In *SAS Users Group International Annual Conference Proceedings*, 2006. URL <http://www2.sas.com/proceedings/sugi31/040-31.pdf>. 8 pp., covers vertical macro arrays, shows many examples of list processing and enclosure of items in list.
 - [3] Ronald J. Fehd. SmryEachVar: A data-review routine for all data sets in a libref. In *SAS Global Forum Annual Conference Proceedings*, 2008. URL <http://www2.sas.com/proceedings/forum2008/003-2008.pdf>. Applications Development, 24 pp.; call execute, data review, data structure, dynamic programming, list processing, parameterized includes, utilities (writattr, writvalu) to repair missing elements in data structure; best contributed paper in ApDev.
 - [4] Ronald J. Fehd. List processing routine CallXinc: Calling parameterized include programs using a data set as list of parameters. In *Western Users of SAS Software Annual Conference Proceedings*, 2009. URL www.lexjansen.com/wuss/2009/app/APP-Fehd2.pdf. Applications Development, 20 pp.; call execute, data review, data structure, dynamic programming, list processing, parameterized includes, examples.
 - [5] Ronald J. Fehd. List processing macro call-macro. In *MidWest SAS Users Group Annual Conference Proceedings*, 2014. URL www.mwsug.org/proceedings/2014/BB/MWSUG-2014-BB04.pdf. Coders Corner, 19 pp.; using %sysfunc with SCL functions to read a list, a control data set, and for each observation, call a macro with variable names and values as named parameters.
 - [6] Ronald J. Fehd and Art Carpenter. List processing basics: Creating and using lists of macro variables. In *SAS Global Forum Annual Conference Proceedings*, 2007. URL <http://www2.sas.com/proceedings/forum2007/113-2007.pdf>. Hands On Workshop, 20 pp.; comparison of methods: making and iterating macro arrays, scanning macro variable, writing calls to macro variable, write to file then include, call execute; using macro function nrstr in call execute argument; 11 examples, bibliography.
 - [7] Russ Lavery. An Animated Guide(C): How %str and %nrstr work. In *NorthEast SAS Users Group Conference Proceedings*, 2004. URL <http://www.lexjansen.com/nesug/nesug04/pm/pm25.pdf>. Programming and Manipulation, 13 pp., thorough explanation of the difference between six types of compile and execution of macro tokens before SAS tokens; discusses functions used to mask special characters; recommended.
 - [8] Shan Lee. Quoting macro variable references. In *Pharmaceutical Users of Software Exchange*, 2007. URL <http://www.lexjansen.com/phuse/2007/tu/TU04.pdf>. Tutorials, 6 pp., comparison of functions %nrstr and %nrbquote with single and double quotes.
 - [9] Robert J. Morris. Text utility macros for manipulating lists of variable names. In *SAS Users Group International Annual Conference Proceedings*, 2005. URL <http://www2.sas.com/proceedings/sugi30/029-30.pdf>. Coders Corner, 7 pp., examples of common problems of quoting items in a list.
 - [10] Mary F. O. Rosenbloom and Art Carpenter. Macro quoting to the rescue: Passing special characters. In *SAS Global Forum Annual Conference Proceedings*, 2013. URL <http://support.sas.com/resources/papers/proceedings13/005-2013.pdf>. Applications Development, 12 pp., storing and retrieving special characters; using macro quoting functions.
 - [11] Kevin Russell. SAS(R) macros: Top ten questions (and answers!). In *MidWest SAS Users Group Annual Conference Proceedings*, 2010. URL <http://www.mwsug.org/proceedings/2010/advanced/MWSUG-2010-174.pdf>. section 39 slides; %sysfunc with data step functions, quoting functions to mask special characters, resolving macro variable references inside single quotes.
 - [12] Jerry S. Tsai. Efficiently handle lists of items using the SAS Macro Language — save time and reduce errors. In *Western Users of SAS Software Annual Conference Proceedings*, 2008. URL <http://www.lexjansen.com/wuss/2008/cod/cod06.pdf>. Coders Corner, 8 pp., use of horizontal and vertical arrays of macro variables in utility macros; best contributed paper.
 - [13] Ian Whitlock. Names, names, names — make me a list. In *SAS Global Forum Annual Conference Proceedings*, 2007. URL <http://www2.sas.com/proceedings/forum2007/052-2007.pdf>. Coders Corner, 10 pp.; binary list operators, use of double quotes and delimiters/separators, uses of macro functions %bquote, %qscan, %str, %superq, %unquote.
-